
Revit API UI実習        

 

    Lab 1 – リボン  

May, 2011 by A. Nagy  

Updated by DevTech AEC WG 

Last modified: 8/4/2015 

 

<VB.NET>VB.NETバージョン</VB.NET>  

目的:この実習では、リボン タブとリボン パネルを作成して、コントロールをそれらに配置する方法を学習します。

学習する項目は次のとおりです。 

         新しいリボン タブおよびパネルを作成する 

         様々なリボン コントロールを作成する 

この実習の実装と確認の手順は、下記のとおりです: 

1.    新しい外部アプリケーションを作成する 

2.    新しいリボン タブおよびパネルを作成する 

3.    プッシュ ボタンを作成する 

4.    スプリット ボタンを作成する 

5.    コンボ ボックスを作成する 

6.    他のコントロールを作成する 

7.    サマリ 

  

1.   新しい外部アプリケーションを定義する 

新しいプロジェクトを作成して、外部アプリケーションを追加します   

1.1  Revit API入門実習の手順に沿ってプロジェクトを作成します。ただし、今回のクラス ライブラリ プロジェクト

の名前は UiVb または UiCs とします(プログラミング言語に依存して変更)。ここでは VB.NET を選択しま

す。 

1.2  Class1.vb ファイルの名前を変更します。 

         ファイル名:  1_Ribbon.vb 

         アプリケーション クラス名:UIRibbon  

  

必要となる参照: 

この実習で必要となる追加の参照は、次のとおりです:   

         PresentationCore － ビットマップ イメージ処理で使用  

         WindowsBase － ビットマップ イメージ処理で使用  

         System.Xaml － ビットマップ イメージ処理で使用  

         IntroCs/Vb － 入門実習で定義したコマンドを使用  

  

  

要求される名前空間: 

この実習で必要とする名前空間は次のとおりです:   

         Autodesk.Revit.DB 

         Autodesk.Revit.UI 

         Autodesk.Revit.ApplicationServices 

         Autodesk.Revit.Attributes 

         System.Collections.Generic 

         System.Xaml 

         System.Diagnostics － デバッグで使用  

         System.IO － フォルダを参照するために使用  

         System.Windows.Media.Imaging － ビットマップ イメージ処理で使用  

1.3  パスを含むいくつかの変数を宣言します(例、リボン コントロール用のイメージ取得のため): 

<VB.NET> 
  ''' <summary> 
  ''' This is both the assembly name and the namespace  
  ''' of the external command provider. 
''' </summary> 
  
  Private Const _dllExtension As String = ".dll" 
  Private Const _introLabName As String = "IntroVb" 
  Private Const _uiLabName As String = "UiVb" 
  
  ''' <summary> 
  ''' Name of subdirectory containing images. 
''' </summary> 
  
  Private Const _imageFolderName As String = "Images" 
  
  ''' <summary> 
  ''' Location of images for icons.  
''' </summary> 
  
  Private _imageFolder As String 
  
  ''' <summary> 
  ''' Location of managed dll where we have defined the commands. 
''' </summary> 
  
  Private _introLabPath As String 

 </VB.NET> 
  

1.4  ユーティリティ関数を 2つ追加します。これらは、イメージ フォルダの検索や特定の画像ファイルから

BitmapImageオブジェクトを作成する手助けをするものです。 

<VB.NET> 
  ''' <summary> 
  ''' Starting at the given directory, search upwards for  
  ''' a subdirectory with the given target name located 
  ''' in some parent directory.  
  ''' </summary> 
  ''' <param name="path">Starting directory, e.g. 

GetDirectoryName( GetExecutingAssembly().Location ).</param> 
  ''' <param name="target">Target subdirectory name, e.g. "Images".</param> 
''' <returns>The full path of the target directory if found, else 

null.</returns> 
  
  Private Function FindFolderInParents( _  
    ByVal path As String, ByVal target As String) As String 
  
    Debug.Assert(Directory.Exists(path), _ 
      "expected an existing directory to start search in") 
    Do 

      Dim s As String = System.IO.Path.Combine(path, target) 
      If Directory.Exists(s) Then 
        Return s 
      End If 
      path = System.IO.Path.GetDirectoryName(path) 
    Loop While (path IsNot Nothing) 
    Return Nothing 
  End Function 
  
  ''' <summary> 
  ''' Load a new icon bitmap from our image folder. 
''' </summary> 
  
  Function NewBitmapImage(ByVal imageName As String) As BitmapImage 
    Return New BitmapImage(New Uri(Path.Combine(Me._imageFolder, imageName))) 
  End Function 
 </VB.NET> 
  

2.   新しいリボン タブおよびパネルを作成する 

 
  

Revit 2012 以降のバージョンでは、CreateRibbonTab()を使用して独自のリボン タブを加えることができます。

使用する CreateRibbonPanel() のオーバロードによって、新しく作成されたパネルは、指定するリボン タブや

既定の [アドイン] タブのいずれかに追加されます。それでは、リボン タブを作成して、そこにパネルを追加する関

数を作成します。 

<VB.NET>  
  Sub AddRibbonSampler(ByVal app As UIControlledApplication) 
  
    ' (1) create a ribbon tab and panel  
  
    app.CreateRibbonTab("Ribbon Sampler") 
  
    Dim panel As RibbonPanel = _  
      app.CreateRibbonPanel("Ribbon Sampler", "Ribbon Sampler") 
  
  End Sub 
 </VB.NET> 
  

リボンの処理を記述する前に、クラス内のOnStartup() 関数の内側でヘルパー変数を初期化していきます。 

  
<VB.NET>  
    ' External application directory: 
  
    Dim dir As String = Path.GetDirectoryName( _ 

        System.Reflection.Assembly.GetExecutingAssembly.Location) 
  
    _introLabPath = Path.Combine(dir, _introLabName + _dllExtension) 
  
    ' External command path: 
  
    If Not File.Exists(_introLabPath) Then 
      TaskDialog.Show( _  
        "UIRibbon", "External command assembly not found: " + _introLabPath) 
      Return Result.Failed 
    End If 
  
    ' Image path: 
  
    _imageFolder = FindFolderInParents(dir, _imageFolderName) 
  
    If _imageFolder Is Nothing Or Not Directory.Exists(_imageFolder) Then 
      TaskDialog.Show( _ 
          "UIRibbon", _ 
          String.Format( _ 
            "No image folder named '{0}' found in the parent directories of 

'{1}.", _ 
            _imageFolderName, dir)) 
      Return Result.Failed 
    End If 
 </VB.NET> 
  

3.   プッシュ ボタンを作成する 

いくつかのコントロールを、上で作成したリボン パネルに追加していきます。最初に、プッシュ ボタンを作成します。

コントロールを作成する際には、まず、作成しようとしているコントロールのプロパティを設定するための、[コントロ

ール名]Data クラスを作成する必要があります。また、実際のコントロールは、AddItem() を呼び出すことで、

作成されることになります(例、コントロール [PushButton] を作成する場合には、 PushButtonData インスタ

ンスを作成してプロパティを設定し、AddItem() の引数に渡します)。  

 

ボタンの場合には、ボタンがクリックされた時に呼び出される外部コマンドの完全なクラス名と、そのクラスを実装

するアセンブリのパスを指定する必要があります。 

 

注意: アドイン マニフェスト ファイル(*.addin) 中で外部コマンドを指定すれば、コマンドは、[アドイン]リボンタブ 

>> [外部] リボンパネル >> [外部ツール] からも利用可能になります。あるいは、様々なリボン コントロールを作

成し、それらのコントロールからコマンドを利用することも可能です。後者の場合、アドイン マニフェスト ファイル内

で外部コマンドを宣言する必要はありません。 

  
<VB.NET>  
  Sub AddPushButton(ByVal panel As RibbonPanel) 
  
    ' Set the information about the command we will be assigning  
    ' to the button  
  
    Dim pushButtonDataHello As _  

      New PushButtonData("PushButtonHello", "Hello World", _  
        _introLabPath, _introLabName + ".HelloWorld") 
  
    ' Add a button to the panel  
  
    Dim pushButtonHello As PushButton = panel.AddItem(pushButtonDataHello) 
  
    ' Add an icon  
    ' Make sure you reference WindowsBase and PresentationCore,  
    ' and import System.Windows.Media.Imaging namespace. 
  
    pushButtonHello.LargeImage = NewBitmapImage("ImgHelloWorld.png") 
  
    ' Add a tooltip 
  
    pushButtonHello.ToolTip = "simple push button" 
  
  End Sub 
 </VB.NET> 
  

4.   スプリット ボタンを作成する 

スプリット ボタンは、基本的にプッシュ ボタンをまとめて表示するボタンです。したがって、このコントロールを作成

するためには、複数の PushButtonData を作成して、それらを SplitButtonData オブジェクトに加える必要が

あります。ここでも、ボタンはそれぞれ特定の外部コマンドに関連付けられます。 

  
<VB.NET>  
  Sub AddSplitButton(ByVal panel As RibbonPanel) 
  
    ' Create three push buttons for split button drop down 
    ' #1  
    Dim pushButtonData1 As _  
      New PushButtonData("SplitCommandData", "Command Data", _  
      _introLabPath, _introLabName + ".CommandData") 
    pushButtonData1.LargeImage = NewBitmapImage("ImgHelloWorld.png") 
  
    ' #2  
    Dim pushButtonData2 As _  
      New PushButtonData("SplitDbElement", "DB Element", _  
      _introLabPath, _introLabName + ".DbElement") 
    pushButtonData2.LargeImage = NewBitmapImage("ImgHelloWorld.png") 
  
    ' #3   
    Dim pushButtonData3 As _  
      New PushButtonData("SplitElementFiltering", "ElementFiltering", _  
      _introLabPath, _introLabName + ".ElementFiltering") 
    pushButtonData3.LargeImage = NewBitmapImage("ImgHelloWorld.png") 
  
    ' Make a split button now  
    Dim splitBtnData As New SplitButtonData("SplitButton", "Split Button") 
    Dim splitBtn As SplitButton = panel.AddItem(splitBtnData) 
    splitBtn.AddPushButton(pushButtonData1) 

    splitBtn.AddPushButton(pushButtonData2) 
    splitBtn.AddPushButton(pushButtonData3) 
  
  End Sub 

 </VB.NET> 
  

5.   コンボ ボックスを作成する 

コンボ ボックスの場合は、コンボ ボックス内の各項目に特定の外部コマンドを関連付けるのではなく、選択イベ

ントで処理します。コンボ ボックスを作成したら、選択項目の変更を処理するために CurrentChanged イベン

トを監視します。 

  
<VB.NET>  
  Sub AddComboBox(ByVal panel As RibbonPanel) 
  
    ' Create five combo box members with two groups  
    ' #1  
    Dim comboBoxMemberData1 As _  
      New ComboBoxMemberData("ComboCommandData", "Command Data") 
    comboBoxMemberData1.Image = NewBitmapImage("Basics.ico") 
    comboBoxMemberData1.GroupName = "DB Basics" 
  
    ' #2  
    Dim comboBoxMemberData2 As _  
      New ComboBoxMemberData("ComboDbElement", "DB Element") 
    comboBoxMemberData2.Image = NewBitmapImage("Basics.ico") 
    comboBoxMemberData2.GroupName = "DB Basics" 
  
    ' #3   
    Dim comboBoxMemberData3 As _  
      New ComboBoxMemberData("ComboElementFiltering", "Filtering") 
    comboBoxMemberData3.Image = NewBitmapImage("Basics.ico") 
    comboBoxMemberData3.GroupName = "DB Basics" 
  
    ' #4 
    Dim comboBoxMemberData4 As _  
      New ComboBoxMemberData("ComboElementModification", "Modify") 
    comboBoxMemberData4.Image = NewBitmapImage("Basics.ico") 
    comboBoxMemberData4.GroupName = "Modeling" 
  
    ' #5 
    Dim comboBoxMemberData5 As _  
      New ComboBoxMemberData("ComboModelCreation", "Create") 
    comboBoxMemberData5.Image = NewBitmapImage("Basics.ico") 
    comboBoxMemberData5.GroupName = "Modeling" 
  
  
    ' Make a combo box now  
    Dim comboBxData As New ComboBoxData("ComboBox") 
    Dim comboBx As ComboBox = panel.AddItem(comboBxData) 
    comboBx.ToolTip = "Select an Option" 
    comboBx.LongDescription = "select a command you want to run" 

    comboBx.AddItem(comboBoxMemberData1) 
    comboBx.AddItem(comboBoxMemberData2) 
    comboBx.AddItem(comboBoxMemberData3) 
    comboBx.AddItem(comboBoxMemberData4) 
    comboBx.AddItem(comboBoxMemberData5) 
  
    AddHandler comboBx.CurrentChanged, New EventHandler(Of _    
      ComboBoxCurrentChangedEventArgs)(AddressOf comboBx_CurrentChanged) 
  
  End Sub 
  
  ''' <summary> 
  ''' Event handler for the above combo box  
''' </summary> 
  
  Sub comboBx_CurrentChanged(ByVal sender As Object, _  
    ByVal e As ComboBoxCurrentChangedEventArgs) 
  
    ' Cast sender as TextBox to retrieve text value 
  
    Dim combodata As ComboBox = TryCast(sender, ComboBox) 
    Dim member As ComboBoxMember = combodata.Current 
    TaskDialog.Show("Combobox Selection", "Your new selection: " + _  
      member.ItemText) 
  
  End Sub 
 </VB.NET> 
  

6.   他のコントロールを作成する 

これまでに紹介したコントロール以外にも様々なコントロールがあります。他のコントロールについては、

Autodesk.Revit.UI 名前空間から辿って見つけることができます。 

 

ここでは、AddRibbonSampler() 関数から、すべてのコントロール生成関数を呼び出します。 

  
</VB.NET>  
  Sub AddRibbonSampler(ByVal app As UIControlledApplication) 
  
    ' (1) create a ribbon tab and panel  
  
    app.CreateRibbonTab("Ribbon Sampler") 
  
    Dim panel As RibbonPanel = _  
      app.CreateRibbonPanel("Ribbon Sampler", "Ribbon Sampler") 
  
    AddPushButton(panel) 
  
    AddSplitButton(panel) 
  
    AddComboBox(panel) 

  End Sub 
 </VB.NET> 
  

次に、OnStartup() の終わりで AddRibbonSampler() を呼び出してください。 

  

また、リボン パネル上にスライドアウト コントロールを置くことも出来ます。 これは、ユーザがリボン パネルの最下

段をクリックすることで表示される部分を指します。 

panel.AddSlideOut() の呼び出し後にすべてのコントロールがパネルに追加されます。 

  

7.   サマリ 

この実習では、リボン タブとリボン パネルを作成して、コントロールをそれらに配置する方法を学習しました。学

習した項目は次のとおりです。 

         新しいリボン タブおよびパネルを作成する 

         様々なリボン コントロールを作成する 

  

Autodesk Developer Network  

  


